
8 KONCEPCJA ZASTOSOWANIA ARKUSZA KALKULACYJNEGO           

DO WSPOMAGANIA REALIZACJI PROCESU DIAGNOSTYKI          

ORAZ NAPRAW W WYBRANYM SZPITALU 

   BARTOSZ SZCZĘŚNIAK, ANNA BUJANOWSKA 

8.1 Wprowadzenie 

Utrzymanie ruchu w obiekcie szpitalnym opiera się na podejmowaniu decyzji eksploata-

cyjnych na podstawie bieżącej oceny stanu technicznego. Wzrost złożoności środków tech-

nicznych i odpowiedzialności za zadania, jakie te środki realizują przyczynił się do powstania 

diagnostyki technicznej, narzędzia kształtującego jakość systemów. Czynnikiem bezpośrednio 

stymulującym rozwój diagnostyki jest odpowiedzialność realizowanej przez obiekt funkcji, 

która może być definiowana w trudno wymiernych kategoriach bezpieczeństwa ludzi, lub też 

w kategoriach ekonomicznych wydajności i efektywności produkcji [3, 9]. Celem tej dziedzi-

ny jest określenie szeroko rozumianego stanu technicznego obiektów za pomocą bezstron-

nych metod i środków dla podwyższenia ich trwałości, niezawodności i efektywności działa-

nia. Znajomość stanu technicznego obiektu, określana przez zbiór istotnych właściwości, nie-

zbędna jest do podejmowania decyzji w każdej fazie życia obiektu, w fazie konstruowania i 

badań prototypu, w fazie wytwarzania oraz w fazie eksploatacji, łącznie z jego likwidacją 

opartą na dobrym rozeznaniu stanu. Badanie diagnostyczne ogólnie składa się z wykrycia 

uszkodzenia bądź odebrania sygnału emitowanego przez obiekt, porównania ze stanem wzor-

cowym bądź wydzielenia w obiekcie charakterystycznych cech i porównania ich z cechami 

sygnałów odpowiadających możliwym stanom obiektu [3, 4, 5, 9, 10, 11]. Patrząc syntetycz-

nie należy wyróżnić trzy dziedziny wiedzy, niezbędnej do prawidłowej oceny stanu obiektu. 

Są to : 

 wiedza o obiekcie badań, 

 wiedza o uszkodzeniach, sygnałach i symptomach, 

 wiedza w zakresie wnioskowania diagnostycznego. 

Wynikiem badania diagnostycznego jest ustalenie stanu obiektu, a następnie podjęcie de-

cyzji co do dalszego użytkowania obiektu i jego składników, aż do następnego planowanego 

terminu badań diagnostycznych [9]. Poprzez różnorodne metody diagnostyczne, możemy 

uzyskać informację o aktualnym stanie obiektu.  

Medyczny sprzęt szpitalny stanowi dosyć zróżnicowaną grupę wyrobów. Jego charakte-

rystykę szczegółowo opisano w [2]. Na tej podstawie do analizy wybrano typ urządzeń bez 

źródeł promieniowania jonizującego oraz przy zakupie których nie zawarto umowy 

o serwisowanie. 

Diagnozowanie sprzętu medycznego w obiekcie szpitalnym nastąpi w wyniku zaistnienia 

dwóch zdarzeń: 

 przeprowadzania zaplanowanego procesu przeglądu, 

 wykrycia uszkodzenia/usterki w trakcie użytkowania - awaria. 

Przegląd jest jednym z procesów eksploatacji infrastruktury technicznej w szpitalu, który 

można zdefiniować jako obszerny zakres działań obsługowych na odpowiednim poziomie dla 

zapewnienia zdatności i bezpieczeństwa obiektu. Wystąpienie zdarzenia awaryjnego urucho-

mi proces naprawy. 

W artykule [6] przestawiono koncepcje wspomagania diagnostyki technicznej w przy-

padku gdy zdarzeniem wszczynającym diagnozowanie jest przegląd. Artykuł jest kontynuacją 

badań autorów i przedstawia narzędzie, które wspiera procesu diagnostyczny w przypadku 

wystąpienia awarii. 

8.2 Opis procesu i założona funkcjonalność narzędzia 

Proces, do wspomagania którego zaproponowane zostanie narzędzie informatyczne roz-

poczyna się w przypadku zauważenia przez pracownika problemu związanego z działaniem 

dowolnego urządzenia w szpitalu. W pierwszej kolejności podejmowana jest decyzja, czy 

diagnostyka problemu może być przeprowadzona przez pracowników szpitala, czy konieczna 

jest usługa serwisu zewnętrznego. W obu przypadkach wstępnie szacowany jest koszt diagno-

styki. Jeżeli koszt przekracza założoną wartość progową konieczne jest utworzenie dokumen-

tu zwanego „protokołem konieczności” który zawiera wykaz przewidywanych kosztów. Jeże-

li protokół konieczności nie zostanie zaakceptowany działania diagnostyczne nie mogą zostać 

podjęte. W przypadku akceptacji protokołu możliwa jest realizacja własnych lub zlecenie ze-

wnętrznych działań diagnostycznych. Po przeprowadzeniu diagnostyki podejmowana jest de-

cyzja o konieczności podjęcia dalszych działań diagnostycznych lub działań naprawczych. W 

przypadku działań naprawczych,  tak jak w przypadku diagnostyki, podejmowana jest decyzja 

o możliwości wykonania naprawy przez pracowników szpitala lub o konieczności zlecenia 

naprawy na firmie zewnętrznej. Również w tym przypadku, w sytuacji gdy przewidywany 

koszt naprawy przekracza założoną wartość, kontynuacja działań wymaga przygotowania 

oraz akceptacji protokołu konieczności. Po zakończeniu naprawy, oceniany jest jej skutek 

oraz podejmowana jest decyzja o kontynuacji działań diagnostycznych lub naprawczych lub 

o zakończeniu prac związanych z danym problemem. 

Do wspomagania procesu koordynowania diagnostyki oraz napraw zaproponowano na-

rzędzie informatyczne zbudowane w oparciu o arkusz kalkulacyjny MS Excel. Założona 

funkcjonalność takiego narzędzia obejmuje: 

 Przechowywanie informacji na temat działań diagnostycznych oraz działań na-

prawczych podejmowanych w związku z wystąpieniem poszczególnych awarii, 

 Wspomaganie wyboru osób którym zostanie powierzona diagnostyka urządzenia, 

poprzez wyświetlenie pracowników szpitala mogących ją  przeprowadzić wraz z 

danymi mogącymi ułatwić kontakt z poszczególnymi osobami, 

 Wspomaganie wyboru osób którym zostanie powierzona naprawa urządzenia, po-

przez wyświetlenie pracowników szpitala mogących ją  przeprowadzić wraz z da-

nymi mogącymi ułatwić kontakt z poszczególnymi osobami, 

 Wspomaganie wyboru firm, którym może zostać zlecone przeprowadzenie działań 

diagnostycznych, poprzez wyświetlenie zestawienia nazw oraz danych kontakto-

wych firm mogących takie działania przeprowadzić , 

 Wspomaganie wyboru firm, mogących przeprowadzić naprawę danego urządzenia, 

poprzez wyświetlenie zestawienia, 

 Umożliwienie przeglądania  informacji na temat wybranej awarii, 

 Wspomaganie generowania „Protokołu konieczności”. 

Założono, że dla zapewnienia możliwości dalszego jego rozwijania przez osoby niepo-

siadające umiejętności programowania a także ze względu bezpieczeństwa, proponowane na-

rzędzie zostanie opracowane przy wykorzystaniu standardowych funkcji oraz wbudowanych 

mechanizmów arkusza.  

8.3 Struktura encji 

Na podstawie przeprowadzonej analizy z uwzględnieniem założonej funkcjonalności na-

rzędzia informatycznego zaproponowano uwzględnienie w nim takich  encji jak „urządzenie”, 

„awaria”, „działanie”, „pracownik”, „firma”, „protokół konieczności”, „pozycja protokołu”, 

„firma-urządzenie”, „pracownik –urządzenie”. Ogólna charakterystyka poszczególnych encji 

została przedstawiona w tabeli 8.1. 

Tabela 8.1. Ogólna charakterystyka zidentyfikowanych encji 

L.p. Nazwa encji Opis 

1 „Urządzenie” Dowolne urządzenie wykorzystywane w szpitalu, w stosunku do którego może 

zajść potrzeba podjęcia działań diagnostycznych lub naprawczych. 

2 „Awaria” Zauważony przez pracownika problem dotyczący dowolnego urządzenia.  

3 „Działanie” Działanie podjęte w celu wyeliminowania zauważonego problemu. 

Podejmowane działania mogą zostać podzielone na działania diagnostyczne 

oraz działania naprawcze.  

4 „Pracownik” Pracownik zatrudniony w szpitalu. Poszczególni pracownicy mogą 

wykonywać diagnostykę oraz naprawy wybranych urządzeń. Niektórzy 

pracownicy mogą podejmować decyzję o podejmowaniu określonych działań 

na podstawie szacunkowych kosztów przedstawianych w protokołach 

konieczności. 

5 „Firma” Firma zewnętrzna, która może przeprowadzić działania diagnostyczne oraz 

działania naprawcze wybranych urządzeń. 

6 „Protokół 

konieczności” 

Dokument, w którym przedstawiane są przewidywane koszty zakładanych 

działań diagnostycznych lub naprawczych. Dokument jest przygotowywany w 

przypadku, gdy przewidywany koszt działania przekracza założony próg 

graniczny. Na podstawie dokumentu pracownik posiadający odpowiednie 

uprawnienia podejmuje decyzję o akceptacji lub braku akceptacji dla 

określonego działania. 

7 „Pozycja protokołu” Pojedyncza pozycja w protokole konieczności, odnosząca się do określonego 

typu wydatków oraz zawierająca jego przewidywaną wysokość. 

8 „Firma-urządzenie” Encja intersekcji identyfikująca powiązanie poszczególnych firm z 

określonymi urządzeniami. Przypisanie firmy do urządzenia w ramach tej 

encji oznacza, że może ona wykonywać jego diagnostykę oraz naprawy.  

9 „Pracownik-

urządzenia” 

Encja intersekcji identyfikująca powiązanie poszczególnych pracowników z 

określonymi urządzeniami. Podobnie jak w przypadku firm, przypisanie 

pracownika do urządzenia w ramach tej encji oznacza, że może on 

wykonywać jego diagnostykę oraz naprawy. 

Tabela 8.2 Charakterystyka oraz nazwy skrócone atrybutów poszczególnych encji 

Nazwa encji Nazwa atrybutu Skrócona 

nazwa 

atrybutu 

Charakterystyka 

„Urządzenie” Kod urządzenia Id_u Kod identyfikujący urządzenie 

Nazwa urządzenia Nazwa Nazwa danego urządzenia 

„Awaria” Kod awarii Id_a Kod identyfikujący zauważony problem 

Data awarii Data Data zauważenia problemu 

Opis Opis Krótki opis charakteryzujący problem 

Koniec Koniec Określa, czy działania podejmowane w związku z 

dana awaria zostały ostatecznie zakończone 

„Działanie” Kod działania Id_d Kod identyfikujący podejmowane działanie 

Rodzaj DN Określa, czy działanie ma charakter diagnostyczny 

czy naprawczy 

Data rozpoczęcia Data_z Data zainicjowania działania 

Podmiot 

wykonujący 

PF Określa czy działanie jest wykonywane przez 

pracowników szpitala, czy przez firmę zewnętrzną 

Wymagana 

akceptacja 

Wym_akc Określa czy dla danego działania konieczna jest 

akceptacja protokołu konieczności 

Termin prac Termin Zaplanowany termin prac 

Data prac Data_d Rzeczywista data rozpoczęcia prac 

Rezultat Rezultat Efekt działania 

„Pracownik” Kod pracownika Id_p Kod identyfikujący pracownika szpitala 

Nazwisko Nazwisko Nazwisko pracownika szpitala 

Imię Imię Imię pracownika szpitala 

Telefon Telefon Telefon kontaktowy pracownika szpitala 

Dyzyzje Decyzje Określa, czy pracownik może akceptować protokół 

konieczności 

„Firma” Kod firmy Id_f Kod firmy mogącej świadczyć usługi w zakresie 

diagnostyki lub napraw 

Nazwa Nazwa Nazwa firmy 

Adres Adres Adres firmy 

Telefon Telefon Telefon kontaktowy firmy 

„Protokół 

konieczności” 

Numer protokołu Numer Numer protokołu konieczności 

Data utworzenia Data_w Data przygotowania protokołu konieczności 

Akceptacja Akceptacja Określa, czy protokół został zaakceptowany 

Data akceptacja Data_d Data podjęcia decyzji 

Tekst nagłówka 1 Nagl_t1 Tekst pierwszego wiersza nagłówka protokołu 

Tekst nagłówka 2 Nagl_t2 Tekst drugiego wiersza nagłówka protokołu 

Tekst nagłówka 3 Nagl_t3 Tekst trzeciego wiersza nagłówka protokołu 

Uzasadnienie Uzasadnieni

e 

Tekst uzasadnienia dla protokołu konieczności 

„Pozycja 

protokołu” 

Pozycja Poz Numer pozycji na protokole konieczności 

 Przedmiot Przedmiot Materiał, element, usługa konieczny do realizacji 

działania, dla którego tworzony jest protokół 

konieczności 

Ilość Ilość Wymagana ilość dla danego przedmiotu 

Cena jednostkowa Cena Cena jednostkowa dla danego przedmiotu 

Stawka VAT Vat Stawka VAT dla danego przedmiotu 

 

W ramach każdej encji zidentyfikowane zostały określone atrybuty. Ich zestawienie oraz 

krótka charakterystyka zostały przedstawione w tabeli 8.2. W tabeli przedstawione zostały 

również nazwy skrócone, które zostały wykorzystane w przedstawionym w dalszej części ar-

tykułu narzędziu.  

Związki pomiędzy poszczególnymi encjami zostały przedstawione na rys. 8.1 za pomocą 

modelu ERD opracowanego zgodnie z koncepcją CASE*Method 0. W przypadku wielokrot-

nych związków pomiędzy encjami w celu uniknięcia niejasności podane zostały także ich na-

zwy 

 

Rys. 8.1 Diagram ERD dla omawianego problemu 

8.4 Reprezentacja encji w arkuszu kalkulacyjnym 

Wszystkie wyszczególnione encje zostały przedstawione w arkuszu kalkulacyjnym w za 

pomocą tabel o układzie identycznym z tabelami w relacyjnej bazie danych. Nagłówki ko-

lumn w tabelach odnoszą się do poszczególnych atrybutów encji, natomiast kolejne wiersze 

zawierają ich kolejne instancje. W odróżnieniu od propozycji [6, 7] reprezentacji encji za po-

mocą tabel będących klasycznym zakresem arkusza w tym przypadku zastosowana została, 

dostępna od wersji MS Excel 2007 [8], struktura arkusza o nazwie „Tabela”. Pozwoliło to na 

uniknięcie konieczności odgórnego ustalania maksymalnej liczby wierszy w przypadku wielu 

tabel. Korzyść ta uzyskiwana jest dzięki: 

 możliwości wykorzystania w wielu formułach odwołań do całego obszaru, który 

został wypełniony danymi, za pomocą nazwy tabeli oraz nazwy atrybutu. 

 automatycznemu przenoszeniu do nowych wierszy istotnych cech poszczególnych 

komórek, w szczególności takich jak wprowadzone formuły, formatowanie oraz 

sprawdzanie poprawności danych. 

Poszczególne tabele znajdują się w oddzielnych arkuszach, każda z nich posiada określo-

ną nazwę: „urządzenie” – arkusz „U” – tabela „TabU”, „awaria” – arkusz „A” – tabela „Ta-

bA”, „działanie” – arkusz „D” – tabela „TabD”, „pracownik” – arkusz „P” – tabela „TabP”, 

„firma” – arkusz „F” – tabela „TabF”,  „protokół konieczności” – arkusz „K” – tabela 

„TabK”, „pozycja protokołu” – arkusz „KP” – tabela „TabKP”, „firma-urządzenie” – arkusz 

„FU” – tabela „TabFU”, „firma-pracownik” – arkusz „FP” – tabela „TabFP”. Nazwy nagłów-

ków w kolumnach tabel są zgodne z nazwami skróconymi atrybutów, które zostały przedsta-

wione w tabeli 8.2. Nazwy nagłówków kolumn, w których znajdują się klucze obce odnoszą-

ce się do określonych związków pomiędzy encjami zostały przedstawione wraz z krótkim wy-

jaśnieniem w tabeli 8.3.  

Tabela 8.3 Wyjaśnienie nazw kolumn zawierających klucze obce 

Tabela Związek Nazwa 

kolumny 

Opis 

TabA 
urządzenie-awaria  Id_u Kod urządzenia, którego dotyczy problem 

pracownik-awaria Id_pz Kod pracownika, który zauważył problem 

TabD 

awaria-działanie Id_a Kod awarii, którego dotyczy działanie 

pracownik-działanie 

firma-działanie 
Id_PF 

Kod pracownika lub firmy, która realizuje określone 

działanie 

TabK 

działanie-protokół 

konieczności 
Id_d 

Kod działanie, dla którego tworzony jest protokół 

konieczności 

pracownik-protokół 

konieczności 
Id_p_w Kod pracownika, który tworzy protokół konieczności 

pracownik-protokół 

konieczności 
Id_p_d 

Kod pracownika, do którego adresowany jest protokół 

konieczności 

TabKP 
protokół konieczności-

pozycja protokołu 
Numer_K 

Numer protokołu, do którego przypisana jest dana 

pozycja 

W „TabD” w celu przedstawienia związku wzajemnie wykluczającego się zastosowano 

metodę wspólnej dziedziny. W polu o nazwie „Id_PF” może być wprowadzony kod firmy lub 

kod pracownika, w zależności od tego, kto wykonuje dane działanie. Wartość wprowadzona 

w polu „PF” określa, jaki rodzaj kodu jest wprowadzany w Poli „Id_PF”.   

8.5 Zapewnienie spójności danych 

Integralność danych w zakresie zgodności kluczy obcych oraz kluczy własnych a także w 

zakresie wyboru założonych wartości dla wybranych atrybutów w poszczególnych tabelach 

można uzyskać w arkuszu MS Excel za pomocą mechanizmu sprawdzania poprawności da-

nych. W ramach powyższego mechanizmu zastosowano opcję pozwalająca na wybieranie do-

puszczalnych wartości z rozwijanych list.  Sprawdzanie poprawności danych zostało zastoso-

wane łącznie z mechanizmem odwoływania się do zakresów arkusza za pomocą ich nazw, 

dzięki czemu możliwe było wykorzystanie wartości pochodzących z innych arkuszy niż ar-

kusz bieżący. W większości przypadków zbiory wartości dopuszczalnych obejmują wszystkie 

wartości wprowadzone w odpowiednich kolumnach powiązanych tabel. W takich sytuacjach 

określając odwołania dla definiowanych nazw zakresu wystarczy podać odwołanie do wybra-

nej kolumny odpowiedniej tabeli. Oprócz wymienionych wcześniej tabel, w narzędziu, w ar-

kuszu S dodana została dodatkowo tabela o nazwie „TabS”, w której wprowadzono wartości 

dopuszczalne dla wybranych argumentów. Wszystkie nazwane obszary, zastosowane w celu 

sprawdzania poprawności danych, łącznie z definicją oraz kolumnami tabel, do których się 

odnoszą zostały przedstawione w tabeli 8.4. W dwóch przypadkach możliwe do wprowadze-

nia wartości te powinny zostać zawężone. Ma to miejsce w tabeli „TabK” w przypadku kodu 

pracownika, do którego adresowany jest protokół konieczności oraz w tabeli „TabD” w przy-

padku kodu firmy lub pracownika realizującego dane działanie. 

W celu rozwiązania problemu związanego z kodem adresata protokołu konieczności, za-

stosowano rozwiązanie wykorzystujące wielokomórkową formułę tablicową, którą wprowa-

dzono w obszarze „K2:K10” w arkuszu „P”. Ten element rozwiązania przedstawiony został 

na rys. 8.2. 

 

FA 
{=JEŻELI.BŁĄD(INDEKS(TabP[Id_p];MIN.K(JEŻELI(TabP[Decyzje]="T"; 

WIERSZ(TabP[Decyzje]);"");WIERSZ(A1:A9))-1;1);"")} 

Rys. 8.2 Element rozwiązania zapewniającego wybór właściwych kodów pracowników,  

do których adresowany jest protokół konieczności  

Przedstawiona formuła w pierwszej kolejności za pomocą funkcji JEŻELI() oraz 

WIERSZ() tworzy jednokolumnową tablicę o liczbie wierszy zgodnej z liczbą wierszy w ta-

beli „TabP”. W każdym  wierszu tablicy wpisywany jest: 

 numer wiersza arkusza, jeżeli są w nim przechowywane dane pracownika posiada-

jącego atrybut „Decyzje” o wartość „T” 

 pusty ciąg znaków, jeżeli w wierszu przechowywane są dane pracownika posiada-

jącego atrybut „Decyzje” o wartość „N” 

W kolejnym kroku utworzona tablica zostaje posortowana za pomocą funkcji MIN.K() 

oraz WIERSZ() tak, żeby w jej początkowych wierszach umieszczone zostały uporządkowane 

w kolejności rosnącej numery wierszy. Wiersze zawierające wartości puste umieszczane są w 

końcowej części tablicy. Na podstawie przygotowanej tablicy za pomocą funkcji INDEKS()z 

kolumny „Id_p” tabeli „TabP” pobierane są kody odpowiednich pracowników. W celu unik-

nięcia pojawiania się błędów jako zewnętrzna funkcje w formule tablicowej zastosowana zo-

stała funkcja JEŻELI.BŁĄD(). Zawierające wartości komórki z obszaru„K2:K10” tworzą 

przedstawiony w tabeli 8.4 zakres o nazwie „Id_p_dec”. 

W celu rozwiązania problemu związanego z kodem pracownika lub firmy wykonującej 

działanie, w arkuszu „U” w każdym wierszu zawierającym dane określonego urządzenia 

utworzono, za pomocą formuł przedstawionych na rys. 8.3 ,zestawienie kodów przypisanych 

do niego pracowników oraz firm.  Również w tym przypadku zastosowane zostały wieloko-

mórkowe formuły tablicowe.  

 

 

B 
{=JEŻELI.BŁĄD(INDEKS(TabPU[Id_p];MIN.K(JEŻELI(TabPU[Id_u]=TabU[[#Ten 

wiersz];[Id_u]];WIERSZ(TabPU[Id_p]);"");NR.KOLUMNY($A$1:$J$1))-1;1);"")} 

FC 
{=JEŻELI.BŁĄD(INDEKS(TabFU[Id_f];MIN.K(JEŻELI(TabFU[Id_u]=TabU[[#Ten 

wiersz];[Id_u]];WIERSZ(TabFU[Id_f]);"");NR.KOLUMNY($A$1:$J$1))-1;1);"")} 

Rys. 8.3 Element rozwiązania zapewniającego wybór właściwych dla danego działania                   

kodów pracowników i firm  

 

W zależności od wartości wprowadzonej w danym wierszu w tabeli „TabD” w kolumnie 

„PF”, odpowiedni dla danego działania ”Id_pf_ogr_D” obejmuje niepuste komórki zawierają-

ce kody pracowników lub firm dopuszczalnych dla urządzenia, którego działanie dotyczy. 

Odpowiednia formuła odwołania dla zakresu „Id_pf_ogr_D” została przedstawiona 

w tabeli 8.4. Aby uniknąć konieczności dalszego rozbudowywania przedstawionej formuły. 

W tabeli „TabD” dodana została kolumna „T_Id_u” zawierająca formułę zwracającą kod 

urządzenia, którego dotyczy awaria, w związku z która podejmowane jest określone działanie. 

Dodatkowa kolumna oraz formuła została przedstawiona na rys. 8.4. 

FD 
=JEŻELI.BŁĄD(INDEKS(TabA[Id_u];PODAJ.POZYCJĘ(TabD[[#Ten 

wiersz];[Id_a]];TabA[Id_a];0);1);"") 

Rys. 8.4 Dodatkowa kolumna oraz formuła zwracająca kod urządzenia,                                                    

którego dotyczy dane działanie  

Tabela 8.4 Nazwane zakresy wykorzystywane w mechanizmie sprawdzania                           

poprawności danych  

Nazwa zakr. Odwołanie Tabela i kolumna 

Id_a =TabA[Id_a] TabD;Id_a 

Id_d =TabD[Id_d] TabK;Id_d 

Id_u =TabU[Id_u] TabA;Id_u 

Numer =TabK[Numer] TabKP;Numer 

Id_p =TabP[Id_p] TabA;Id_pz 

TabK;Id_p_w 

Id_p_dec =PRZESUNIĘCIE(P!$K$1;1;0;9-LICZ.PUSTE(P!$K$2:$K$10);1) TabK;Id_p_d 

Id_pf_ogr_D =JEŻELI(D!$E4="F"; 

      PRZESUNIĘCIE(U!$P$1;PODAJ.POZYCJĘ(TabD[[#Ten wiersz]; 

      [T_Id_u]];TabU[Id_u];0);0;1; 10-LICZ.PUSTE 

(PRZESUNIĘCIE(U!$P$1; 

      PODAJ.POZYCJĘ(TabD[[#Ten wiersz];[T_Id_u]]; 

TabU[Id_u];0);0;1;10))) 

; 

      PRZESUNIĘCIE(U!$F$1;PODAJ.POZYCJĘ(TabD[[#Ten wiersz]; 

     [T_Id_u]];TabU[Id_u];0);0;1; 10-

LICZ.PUSTE(PRZESUNIĘCIE(U!$F$1; 

     PODAJ.POZYCJĘ(TabD[[#Ten wiersz];[T_Id_u]]; 

TabU[Id_u];0);0;1;10))) 

) 

TabD;Id_PF 

Akceptacja =TabS[Akceptacja] TabK;Akceptacja 

Decyzje =TabS[Decyzje] TabP;Decyzje 

DN =TabS[DN] TabD;DN 

Koniec =TabS[Koniec] TabA;Koniec 

Nagl_t1 =TabS[Nagl_t1] TabK;Nagl_t1 

Nagl_t2 =TabS[Nagl_t2] TabK;Nagl_t2 

Nagl_t3 =TabS[Nagl_t3] TabK;Nagl_t3 

PF =TabS[PF] TabD;PF 

Wym_akc =TabS[Wym_akc] TabD;Wym_akc 

8.6 Przeglądanie informacji na temat wybranej awarii 

 

 

FE =JEŻELI.BŁĄD(INDEKS(TabA[Data];PODAJ.POZYCJĘ(A2;TabA[Id_a];0);1);"") 

FF =JEŻELI.BŁĄD(INDEKS(TabU[Nazwa];PODAJ.POZYCJĘ(M4;TabU[Id_u];0);1);"") 

FG 
=JEŻELI.BŁĄD(INDEKS(TabP[Imię];PODAJ.POZYCJĘ(N4;TabP[Id_p];0);1) & " 

"&INDEKS(TabP[Nazwisko];PODAJ.POZYCJĘ(N4;TabP[Id_p];0);1);"") 

FH 
{=JEŻELI.BŁĄD(INDEKS(TabD[Id_d];MIN.K(JEŻELI(TabD[Id_a]=A2;WIERSZ(TabD[Id_a]);""); 

WIERSZ(A1:A4))-1;1);"")} 

FI =JEŻELI.BŁĄD(INDEKS(TabD[PF];PODAJ.POZYCJĘ($A8;TabD[Id_d];0);1);"") 

FJ 

=JEŻELI.BŁĄD(JEŻELI(D8="P";INDEKS(TabP[Imię];PODAJ.POZYCJĘ(M8;TabP[Id_p];0);1)&" "& 

INDEKS(TabP[Nazwisko]; PODAJ.POZYCJĘ(M8;TabP[Id_p];0);1);INDEKS(TabF[Nazwa]; 

PODAJ.POZYCJĘ(M8;TabF[Id_f];0);1));"") 

FK 
{=JEŻELI.BŁĄD(INDEKS(TabPU[Id_p];MIN.K(JEŻELI(TabPU[Id_u]=$M$4; 

WIERSZ(TabPU[Id_u]);"");WIERSZ(A1:A3))-1;1);"")} 

FL =JEŻELI.BŁĄD(INDEKS(TabP[Imię];PODAJ.POZYCJĘ($A15;TabP[Id_p];0);1);"") 

FM 
{=JEŻELI.BŁĄD(INDEKS(TabFU[Id_f];MIN.K(JEŻELI(TabFU[Id_u]=$M$4;WIERSZ(TabFU[Id_u]);"

");WIERSZ(A1:A3))-1;1);"")} 

FN =JEŻELI.BŁĄD(INDEKS(TabF[Nazwa];PODAJ.POZYCJĘ($A21;TabF[Id_f];0);1);"") 

Rys. 8.5 Układ komórek oraz formuły w arkuszu „Awaria” 

 

W celu zapewnienia możliwości przeglądania zebranych informacji na temat poszczegól-

nych awarii utworzony został arkusz „Awaria”. W arkuszu tym po wybraniu z listy kodu 

określonej awarii wyświetlane są wszystkie przechowywane na jej temat informacje. Prezen-

towane informacje można podzielić na cztery grupy: informacje bezpośrednio związane z 

awarią, zestawienie wszystkich podejmowanych w związku z nią działań, wykaz pracowni-

ków firmy oraz wykaz firm zewnętrznych mogących diagnozować oraz naprawiać urządze-

nie, którego dotyczy awaria. Układ komórek w arkuszu „Awaria” wraz z najistotniejszymi 

formułami został przedstawiony na rys. 8.5. Przedstawione na rysunku formuły tablicowe ma-

ją taką samą postać we wszystkich zaznaczonych komórkach. Pozostałe formuły zostały 

przedstawione w postaci odpowiadającej górnej komórce danej kolumny.  

 

 

FO =INDEKS(TabA[Id_u];PODAJ.POZYCJĘ(Awaria!A2;TabA[Id_a];0);1) 

FP =INDEKS(TabA[Id_pz];PODAJ.POZYCJĘ(Awaria!A2;TabA[Id_a];0);1) 

FQ =JEŻELI.BŁĄD(INDEKS(TabD[Id_PF];PODAJ.POZYCJĘ($A8;TabD[Id_d];0);1);"") 

Rys. 8.6 Dodatkowe pola w arkuszu „Awaria” wraz z zawartymi w nich formułami 

Tabela 8.5 Nazwane zakresy uwzględniające pole „Wybór” z arkusza „Awaria” 

Nazwa  

zakresu 

Odwołanie Tabela 

 i kolumna 

Id_pf_ogr_D_A 

=JEŻELI(D!$E17="F"; 

        JEŻELI(LICZ.JEŻELI(Awaria!$E$21:$E$23;1)=0; 

               PRZESUNIĘCIE(U!$P$1;PODAJ.POZYCJĘ(TabD[[#Ten wiersz];[T_Id_u]]; 

               TabU[Id_u];0);0;1;10-LICZ.PUSTE(PRZESUNIĘCIE(U!$P$1; 

               PODAJ.POZYCJĘ(TabD[[#Ten wiersz]; [T_Id_u]];TabU[Id_u];0);0;1;10))) 

               ; 

           INDEKS(Awaria!$A$21:$A$23;PODAJ.POZYCJĘ(1;Awaria!$E$21:$E$23;0);1) 

               ) 

        ; 

               JEŻELI(LICZ.JEŻELI(Awaria!$E$15:$E$17;1)=0; 

               PRZESUNIĘCIE(U!$F$1;PODAJ.POZYCJĘ(TabD[[#Ten wiersz];[T_Id_u]]; 

               TabU[Id_u];0);0;1;10-LICZ.PUSTE(PRZESUNIĘCIE(U!$F$1; 

               PODAJ.POZYCJĘ(TabD[[#Ten wiersz];[T_Id_u]];TabU[Id_u];0);0;1;10))) 

               ; 

           INDEKS(Awaria!$A$15:$A$17;PODAJ.POZYCJĘ(1;Awaria!$E$15:$E$17;0);1) 

               ) 

        ) 

TabD;Id_PF 

Id_d_ogr_K_A 

=JEŻELI(LICZ.JEŻELI(Awaria!$J$8:$J$11;1)=0; 

          TabD[Id_d] 

; 

          INDEKS(Awaria!$A$8:$A$11;PODAJ.POZYCJĘ(1;Awaria!$J$8:$J$11;0)) 

) 

TabK;Id_d 

 

Na rysunku pominięte zostały formuły, różniące się od zaprezentowanych formuł jedynie 

odwołaniem do innej kolumny tabeli. W celu uniknięcia nadmiernego rozbudowywania nie-

których formuł w arkuszu awaria dodane zostały trzy pola dodatkowe. Pola te wraz z zawar-

tymi w nich formułami zostały przedstawione na rys. 8.6. 

W arkuszu „Awaria” w wierszach zawierających informacje o działaniach, o pracowni-

kach oraz o firmach wprowadzono dodatkowe pole „Wybór”. Pole to ma za zadanie ułatwiać 

wprowadzania kodów działań w tabeli „TabK” a także kodów pracowników oraz kodów firm 

w tabeli „TabD”. Po wpisaniu w polu „Wybór” wartości „1” zbiory dopuszczalnych wartości 

dla wymienionych kodów zostaną ograniczone do pojedynczych wartości, co zmniejszy nie-

bezpieczeństwo popełnienia błędu podczas wprowadzania danych. Powyższe rozwiązanie 

wymaga modyfikacji formuły definiującej obszar zawierający dopuszczalne wartości.  Zmo-

dyfikowane dane zostały przedstawione w tabeli 8.5. 

8.7 Protokoły konieczności 

 

 

FR =INDEKS(TabK[Data_w];PODAJ.POZYCJĘ(ProtK!C2;TabK[Numer];0);1) 

FS 

=INDEKS(TabK[Nagl_t1];PODAJ.POZYCJĘ(ProtK!C2;TabK[Numer];0);1) & "  " & 

INDEKS(TabP[Imię];PODAJ.POZYCJĘ(INDEKS(TabK[Id_p_d]; 

PODAJ.POZYCJĘ(ProtK!C2;TabK[Numer];0);1);TabP[Id_p];0);1)& " " & 

INDEKS(TabP[Nazwisko];PODAJ.POZYCJĘ(INDEKS(TabK[Id_p_d]; 

PODAJ.POZYCJĘ(ProtK!C2;TabK[Numer];0);1); TabP[Id_p];0);1) 

FT =INDEKS(TabK[Nagl_t2];PODAJ.POZYCJĘ(ProtK!C2;TabK[Numer];0);1) 

FU =INDEKS(TabK[Nagl_t3];PODAJ.POZYCJĘ(ProtK!C2;TabK[Numer];0);1) 

FW 
=JEŻELI.BŁĄD(INDEKS(TabKP[Poz];PODAJ.POZYCJĘ(ZŁĄCZ.TEKSTY($C$2;"-

";WIERSZ(A1:A5)); TabKP[T_Kod];0);1);"") 

FX 
=JEŻELI.BŁĄD(INDEKS(TabKP[Przedmiot];PODAJ.POZYCJĘ(ZŁĄCZ.TEKSTY($C$2;"-

";WIERSZ(A1:A5)); TabKP[T_Kod];0);1);"") 

FY =INDEKS(TabK[Uzasadnienie];PODAJ.POZYCJĘ(ProtK!C2;TabK[Numer];0);1) 

FZ 

=INDEKS(TabP[Imię];PODAJ.POZYCJĘ(INDEKS(TabK[Id_p_w];PODAJ.POZYCJĘ(ProtK!C2;TabK[

Numer];0);1);TabP[Id_p];0);1) &" "& INDEKS(TabP[Nazwisko];PODAJ.POZYCJĘ( 

INDEKS(TabK[Id_p_w];PODAJ.POZYCJĘ(ProtK!C2;TabK[Numer];0);1);TabP[Id_p];0);1) 

Rys. 8.7. Układ komórek oraz najistotniejsze formuły w arkuszu „ProtK” 

W celu zapewnienia możliwości przeglądania oraz drukowania wybranych protokołów 

konieczności utworzony został arkusz „ProtK”. W arkuszu tym generowany jest protokół ko-

nieczności o wybranym z listy numerze. Układ komórek w arkuszu „ProtK” łącznie z naj-

istotniejszymi formułami pobierającymi dane z odpowiednich tabel zostały przedstawione na 

rys. 8.7. Podobnie jak miało to miejsce w przypadku arkusza „Awarie”, na rys. 8.7 pominięto 

formuły podobne do formuł prezentowanych. Pominięte zostały również formuły wykonujące 

proste działania arytmetyczne. 

8.8 Podsumowanie 

Zaprezentowane rozwiązanie pokazuje, ze arkusz kalkulacyjny może z powodzeniem zo-

stać zastosowany do wspomagania procesu diagnostyki oraz napraw w szpitalu. Rozwiązania 

wykorzystujące wbudowane funkcje oraz mechanizmy arkusza kalkulacyjnego pozwoliły na 

zapewnienie wszystkich założonych funkcjonalności. Przechowywanie danych w arkuszu 

kalkulacyjnym może być zrealizowane za pomocą struktury charakterystycznej dla relacyjnej 

bazy danych. Dostępny w arkuszu mechanizm sprawdzania poprawności danych pozwala na 

zapewnienie integracji danych przechowywanych w poszczególnych tabelach. Dostępne w 

arkuszu wbudowane funkcję pozwalają na tworzenie formuł umożliwiających pobieranie z 

tabel wszystkich potrzebnych danych.   

